John Guckenheimer Philip Holmes Page 134, center manifold reduction for parametrized families of system Duffing's equation

```
> restart:with(LinearAlgebra):
> # dot(u)=f1; dot(v)=f2; and beta is the bifurcation parameter
> f1:= v; f2:= bet*u-u^2 - del*v;
```

$$f1 := v$$

$$f2 := bet\,u - del\,v - u^2 \quad (1)$$

```
> # equilibrium
> soln_fp:= solve({f1,f2},{u,v});
```

$$soln\_fp := \{u = 0, v = 0\}, \{u = bet, v = 0\} \quad (2)$$

```
> bet0:= 0:
> u0:=0: v0:=0:
```

```
> J:= Matrix(2,2,[[diff(f1,u),diff(f1,v)],[diff(f2,u),diff(f2,v)]])
  ;
```

$$J := \begin{bmatrix} 0 & 1 \\ bet - 2\,u & -del \end{bmatrix} \quad (3)$$

```
> J0:= simplify(subs(u=u0,v=v0,bet=bet0,J));
```

$$J0 := \begin{bmatrix} 0 & 1 \\ 0 & -del \end{bmatrix} \quad (4)$$

```
> ev:= Eigenvectors(J0);
```

$$ev := \begin{bmatrix} -del \\ 0 \end{bmatrix}, \begin{bmatrix} -\frac{1}{del} & 1 \\ 1 & 0 \end{bmatrix} \quad (5)$$

```
> lam1:= ev[1][1]; lam2:= ev[1][2];
```

$$lam1 := -del$$

$$lam2 := 0 \quad (6)$$

```
> v1:= Vector(2,[1,0]);#v1:= ev[2][1..2,1];
```

$$v1 := \begin{bmatrix} 1 \\ 0 \end{bmatrix} \quad (7)$$

```
> v2:=Vector(2,[1,-del]);#v2:= ev[2][1..2,2];
```

$$v2 := \begin{bmatrix} 1 \\ -del \end{bmatrix} \quad (8)$$

```
> T:= <v1|v2>;
```

$$T := \begin{bmatrix} 1 & 1 \\ 0 & -del \end{bmatrix} \quad (9)$$

```
> TI:= MatrixInverse(T);
```

$$TI := \begin{bmatrix} 1 & \frac{1}{del} \\ 0 & -\frac{1}{del} \end{bmatrix} \quad \textbf{(10)}$$

```
> TestT:= MatrixMatrixMultiply(TI,MatrixMatrixMultiply(J0,T));
```

$$TestT := \begin{bmatrix} 0 & 0 \\ 0 & -del \end{bmatrix} \quad \textbf{(11)}$$

```
> # new variables Y=(x,y), Old variables X= (u,v)
> X:= Vector(2,[u,v]);
```

$$X := \begin{bmatrix} u \\ v \end{bmatrix} \quad \textbf{(12)}$$

```
> Y:= Vector(2,[x,y]);
```

$$Y := \begin{bmatrix} x \\ y \end{bmatrix} \quad \textbf{(13)}$$

```
> X2:= MatrixVectorMultiply(T,Y);
```

$$X2 := \begin{bmatrix} x+y \\ -del\, y \end{bmatrix} \quad \textbf{(14)}$$

```
> # dot(x)=g1=dx/dt; dot(y)=g2=dy/dt; G= [g1,g2]
> G:= simplify(subs(u=X2[1],v=X2[2],MatrixVectorMultiply(TI,Vector
  (2,[f1,f2]))));
```

$$G := \begin{bmatrix} -\frac{(x+y)\,(x+y-bet)}{del} \\ \frac{y^2+(-del^2-bet+2\,x)\,y+x\,(x-bet)}{del} \end{bmatrix} \quad \textbf{(15)}$$

```
> g1:= expand(G[1]);g2:=expand(G[2]);
```

$$g1 := \frac{bet\,x}{del}+\frac{bet\,y}{del}-\frac{x^2}{del}-\frac{2\,x\,y}{del}-\frac{y^2}{del}$$

$$g2 := -del\,y-\frac{bet\,x}{del}-\frac{bet\,y}{del}+\frac{x^2}{del}+\frac{2\,x\,y}{del}+\frac{y^2}{del} \quad \textbf{(16)}$$

```
> g1L:= subs(x=0,y=0,g1)+subs(x=0,y=0,diff(g1,x))*x+subs(x=0,y=0,
  diff(g1,y))*y;
```

$$g1L := \frac{bet\,x}{del}+\frac{bet\,y}{del} \quad \textbf{(17)}$$

```
> g2L:= subs(x=0,y=0,g2)+subs(x=0,y=0,diff(g2,x))*x+subs(x=0,y=0,
  diff(g2,y))*y;
```

(18)

$$g2L := -\frac{bet\,x}{del} + \left(-del - \frac{bet}{del}\right) y \quad (18)$$

```
> g1nL:= simplify(g1-g1L); g2nL:= simplify(g2-g2L);
```

$$g1nL := -\frac{(x+y)^2}{del}$$

$$g2nL := \frac{(x+y)^2}{del} \quad (19)$$

```
> # for (3.2.32) dot(beta)=0=dbet/dt
> g3:=0;
```

$$g3 := 0 \quad (20)$$

```
> ## seek a center manifold y=h(x,bet)= a*x^2+b*x*bet+c*bet^2 + h.
  o.t.
> h:= a*x^2+b*x*bet+c*bet^2;
```

$$h := a\,x^2 + b\,x\,bet + c\,bet^2 \quad (21)$$

```
> dxh:= diff(h,x); # dy/dx
```

$$dxh := 2\,a\,x + b\,bet \quad (22)$$

```
> dbeth:= diff(h,bet); # dy/dbet
```

$$dbeth := b\,x + 2\,bet\,c \quad (23)$$

```
> N:= sort(expand(subs(y=h,dxh*g1+dbeth*g3-g2)),[x,bet]);
```

$$N := -\frac{2\,a^3\,x^5}{del} - \frac{5\,a^2\,b\,x^4\,bet}{del} - \frac{4\,a\,b^2\,x^3\,bet^2}{del} - \frac{4\,a^2\,c\,x^3\,bet^2}{del} - \frac{6\,a\,c\,b\,x^2\,bet^3}{del} - \frac{b^3\,x^2\,bet^3}{del} - \frac{2\,b^2\,c\,x\,bet^4}{del} - \frac{2\,c^2\,a\,x\,bet^4}{del} - \frac{c^2\,b\,bet^5}{del} - \frac{5\,a^2\,x^4}{del} + \frac{2\,a^2\,x^3\,bet}{del} - \frac{8\,a\,b\,x^3\,bet}{del} + \frac{3\,a\,b\,x^2\,bet^2}{del} - \frac{3\,b^2\,x^2\,bet^2}{del} - \frac{6\,a\,c\,x^2\,bet^2}{del} - \frac{4\,b\,c\,x\,bet^3}{del} + \frac{2\,c\,a\,x\,bet^3}{del} + \frac{b^2\,x\,bet^3}{del} - \frac{c^2\,bet^4}{del} + \frac{c\,b\,bet^4}{del} - \frac{4\,a\,x^3}{del} + \frac{3\,a\,x^2\,bet}{del} - \frac{3\,b\,x^2\,bet}{del} + \frac{2\,b\,x\,bet^2}{del} - \frac{2\,c\,x\,bet^2}{del} + \frac{c\,bet^3}{del} - \frac{x^2}{del} + a\,del\,x^2 + \frac{x\,bet}{del} + b\,del\,x\,bet + c\,del\,bet^2 \quad (24)$$

```
> N_x_2:= subs(x=0,bet=0,diff(N,x$2)/2);
```

$$N\_x\_2 := -\frac{1}{del} + a\,del \quad (25)$$

```
> N_x_bet:= subs(x=0,bet=0,diff(N,x,bet));
```

$$N\_x\_bet := \frac{1}{del} + b\,del \quad (26)$$

```
> N_bet_2:= subs(x=0,bet=0,diff(N,bet$2)/2);
```

$$N\_bet\_2 := c\,del \quad (27)$$

```
> a:= solve(N_x_2,a); b:=solve(N_x_bet,b); c:= solve(N_bet_2,c);
```

$$a := \frac{1}{del^2}$$

$$b := -\frac{1}{del^2}$$

$$c := 0 \qquad (28)$$

```
> gg1:= collect(expand(subs(y=h,g1)),x);
```

$$gg1 := -\frac{x^4}{del^5} + \left(-\frac{2}{del^3} + \frac{2\,bet}{del^5}\right)x^3 + \left(\frac{3\,bet}{del^3} - \frac{1}{del} - \frac{bet^2}{del^5}\right)x^2 + \left(\frac{bet}{del} - \frac{bet^2}{del^3}\right)x \qquad (29)$$